
A Short Introduction to WinBUGS

Ciprian Crainiceanu

WinBUGS = Bayesian analysis software Using Gibbs Sampling for Windows.
It can be downloaded for free from

http://www.mrc-bsu.cam.ac.uk/bugs/winbugs/contents.shtml

Do not forget to download the key for unrestricted use. It is on the same
page. The current version is WinBUGS 1.4. This course uses heavily the WinBUGS
manual, which is recommended together with the large number of examples.

1 Why Bayesian analysis using MCMC is good for you

1. Has a very solid decision-theoretical framework

2. It is intuitive:

a. Combines the prior distribution (prior beliefs and/or experience) with the
likelihood (experiment) to obtain the posterior distribution (accumulated
information).

b. In this context knowledge is equivalent to distribution and knowledge pre-
cision can be quantified by the precision parameter.

3. It is exact. Asymptotic approximations are not used. The “plug–in” principle
is avoided.

4. Newly developed numerical methods (MCMC) make computations tractable for
practically all models.

5. Focus shifts from model estimation to model correctness.

6. WinBUGS

2 MCMC methods

a. Suppose that we have a model for which we know the full joint distribution of
all quantities.

b. We want to sample values of the unknown parameters from their conditional
(posterior) given the data (observed values)

c. Gibbs sampling’s basic idea: successively sample from the conditional distribu-
tion of each variables given all the other.

d. Metropolis-within-Gibbs algorithm: appropriate for difficult full conditionals
and does not necessarily generate a new value at each iteration

e. Under broad conditions this process eventually provides samples from the joint
posterior distribution of the unknown variables.

MCMC simulations contain an adaptive or burn-in period which needs to be dis-
carded. The actual correlated samples from the target distribution are used
for inference using simple summary statistics such as posterior mean, posterior
quantiles, etc.

3 WinBUGS program structure

Every WinBUGS program has three main parts: the actual model, the data and the
initial values. All these can be contained in the same or in separate files. The files
can be text (.txt) and/or WinBUGS specific files (.odc).

I personally favor the (.odc) file for the excellent fold capability. This allows
one to write code, data or initial values between two arrows and then collapse the
entire folder by clicking on one of the arrows. To get an idea, open a new file in
WinBUGS (Menu->File->New) and create a new folder using the following steps
(Menu->Tools->Create Fold). You can now write between the two arrows and then
collapse the folder. This is especially good when multiple models, data sets, text etc.
are stored in the same file.

The standard WinBUGS program stored in one file looks like

model

#This is a comment

{ #begin model

...

} #end model

#Data here in one of the two formats

#Initial parameter values here

Writing in the BUGS file follows the same rules as in, say, Notepad or Word. We
will present a few examples to illustrate WinBUGS programming.

Of course, separate files could also be used for model, data and initial values.
This is actually required if one wants to use the scripting capability of WinBUGS
described in section 6.

Why do you think that folding is helpful?

4 Examples

Generally, real data have a very complex structure which is explained by complex
models. Bayesian model specification (and WinBUGS) takes advantage very often of
simple conditional relationship to build such complex models.

We discuss now three examples for the implementation of statistical models in
WinBUGS.

4.1 The line example

Consider a set of 5 observed (x, Y ) pairs (1, 1), (2, 3), (3, 3), (4, 3), (5, 5). We want
to fit the following model in WinBUGS

{
Yi ∼ NORMAL(µi, τ)
µi = α + β(xi − x̄)

, (1)

where x̄ is the mean of x’s. Here τ is the precision of the normal distribution τ = σ−2.
To complete the specification of the model α, β and τ are given proper but mini-

mally informative priors

{
α, β ∼ NORMAL(0, 10−6)

τ ∼ GAMMA(10−3, 10−3)
, (2)

where the mean and variance of GAMMA(a, b) distribution are a/b = 1 and a/b2 =
103 respectively.

The BUGS language allows a concise expression of the model. Open a new file in
WinBUGS and write the following code

model

{for (i in 1:N)

{Y[i]~dnorm(mu[i],tau)

mu[i]<-alpha+beta*(x[i]-mean(x[]))}

3



alpha~dnorm(0,1.0E-6)

beta~dnorm(0,1.0E-6)

tau~dgamma(1.0E-3,1.0E-3)

sigma<-1/sqrt(tau)

}

Note that the BUGS code follows literally the model specification. Open the Model
menu and open Specification. Highlight the keyword model at the beginning of the
model description by double-clicking on it. Execute check model from the specification
tool. “model is syntactically correct” should appear in the status line at the bottom
left corner of the screen.

Data can be loaded in two formats. A combination of the two formats can also
be used. The data can be represented using S-plus object notation

list(x=c(1,2,3,4,5), Y=c(1,3,3,3,5), N=5)

or as

list(N=5)

x[] Y[]

1 1

2 3

3 3

4 3

5 5

END

To load the data using format 1, highlight the key word list at the start of data
description and then execute the load data command from the specification tool.
“data loaded” should appear in the status line. To load the data using format 2,
highlight the key word list and execute load data, then highlight the x at the start of
the rectangular array and again execute load data.

The second format is especially nice when data is stored in large rectangular arrays
like is the case with data in Excel or Access data bases.

Warning: Do not use Copy/Paste directly from Excel because Win-
BUGS does not understand the spacing. Do Copy/Paste twice going from
Excel->Notepad->WinBUGS.

Warning: Do not forget the keyword END at the end of the rectangular
array. This is specific for WinBUGS 1.4 and was not used in the previous
versions

The MCMC sampler must be given initial values that are consistent with every
implicit or explicit constraint

list(alpha=0,beta=0,tau=1)

Highlight the word list and then execute the load inits command from the Specifi-
cation tool. “initial values loaded: model initialized” should appear in the status
line.

WinBUGS is now ready to generate samples. From the Model menu open the
Update tool, and from the Inference menu open the Samples tool.

Type the name of each node to be monitored in the node box and click on set
after each name. You can adjust the number of simulations in the Update. Click
on the update button of the Update tool and wait until the simulation is over. The
time needed for simulation is displayed in the left corner of the window.

To look at the output of all parameters at once type “*” in the node box of
the Samples tool. This tool offers summary statistics, traces, kernel density plots,
autocorrelation functions, and running quantile estimates. Clicking on coda provides
all the simulated values in two files. This values can be used for more sophisticated
analyses.

Assessing the convergence of chains can be done using CODA. However, the most
practical way of assessing convergence towards the target distribution is to run multi-
ple chains with various initial values and visually inspect the histories of these chains.

4.2 Stacks: robust and ridge regression

Birkes and Dodge (1993) apply different regression models to the stack-loss data of
Brownlee (1965). This features 21 daily responses of stack loss (y), the amount of
ammonia escaping, with covariates being air flow (x1), temperature (x2) and acid
concentration x3. Data looks like this

Day Stack loss air flow temp acid

1 42 80 27 89

2 37 80 27 88

... ... ... ... ...

21 15 70 20 91

We first assume a linear regression with a variety of error structures





µi = β0 + β1z1i + β2z2i + β3z3i

yi ∼ NORMAL(µi, τ)
yi ∼ DOUBLEEXP(µi, τ)
yi ∼ LOGISTIC(µi, τ)
yi ∼ t(µi, τ, d)

, (3)

where

zij =
xij − x̄.j

sd(x.j)

are the covariates standardized to have zero mean and unit variance.

Maximum likelihood estimates for the double exponential (Laplace) distribution
minimize the sum of absolute deviations. The other options are alternative heavy
tailed distributions. We use a t with four degrees of freedom, but with more data it
would be possible to allow this parameter to be unknown.

Initially we allow β1, β2 and β3 to have independent “noninformative” priors. We
also consider “ridge regression”, intended to avoid the instability due to correlated co-
variates. This is equivalent to assuming the regression coefficients of the standardized
covariates to be exchangeable and

βj ∼ NORMAL(0, φ), j = 1, 2, 3.

We can apply ridge regression to each of the possible error distributions. The
following model incorporates all models described here

model{

#Standardize x’s

for (j in 1:p)

{for (i in 1:n)

{z[i,j]<-(x[i,j]-mean(x[,]))/sd(x[,j])}}

#Likelihood definition

d<-4 #degrees of freedom for t

for (i in 1:N){

#One and only one of the next four code lines is not

#commented

# Y[i]~dnorm(mu[i],tau);

# Y[i]~ddexp(mu[i],tau);

Y[i]~dlogis(mu[i],tau);

# Y[i]~dt(mu[i],tau,d);

mu[i]<-beta0+beta[1]*z[i,1]+beta[2]*z[i,2]+beta[3]*z[i,3]}

#Priors

beta0~dnorm(0,1.0E-6)

for (j in 1:p){

# beta[j]~dnorm(0,1.0E-6) #independent coefficients

beta[j]~dnorm(0,phi)} #exchangeable coefficients

tau~dgamma(1.0E-3,1.0E-3)

phi~dgamma(1.0E-3,1.0E-3)

#standard deviation of error distribution

#sigma<-sqrt(1/tau) #normal errors

#sigma<-sqrt(2)/tau #double exponential errors

sigma<-sqrt(pow(PI,2)/3)/tau #logistic errors

#sigma<-sqrt(d/(tau*(d-2))) #t errors on d d.f.

}

4.3 Pathogen counts: Poisson regression

Craniceanu et al., 2003 present a Bayesian model for the ICR survey of national
waterborne pathogen concentrations. The data was obtained by conducting an 18
month long survey that included 350 major water users. The basic data set for this
analysis is the number of pathogens counted.

Consider a given water sample of volume V taken from a natural water body at
a given time. Denote by C the unobserved pathogen concentration. One expects to
have on average N = C×V pathogens in the volume of water sampled. However, the
counting process of microscopic organisms is imperfect and the expected number of
pathogens actually counted is only a fraction, R, of the total number N . R is called
the recovery rate (for simplicity it will be considered constant and known). Thus the
expected number of organisms counted is R× V × C.

The quantity of concern is C which is unobserved while the outcome of the count-
ing process is the number of pathogens observed.

A variety of factors could have a causal relationship with pathogen concentra-
tions such as water quality (turbidity, temperature), basin characteristics (land usage,
standardized flow or residence time, type of water body), seasonality, and spatial and
temporal correlations.

Bayesian model





Yij ∼ POISSON(λij)
λij = VijRijCij

log(Cij) = XT
ijβ + tij

tij|si, σt ∼ NORMAL(si, σ
2
t )

si|rk(i), σs ∼ NORMAL(rk(i), σ
2
s)

rk|µ, σr ∼ NORMAL(µ, σ2
r)

, (4)

Here Yij is the observed pathogen count for site i and month j. The site number
i runs from 1 to M , where M is the number of sites considered. The month index j
runs from 1 to Ni, where generally Ni = 18, but sometimes Ni < 18. Conditional on
their means, the random time-site effects tij, site effects si and regional effects rk are
independent and normally distributed.

In order to write the WinBUGS program I used an indexing procedure that can
be useful in many applications. First, I stacked the vectors of observations for each
site and formed vectors of responses and covariates. A one-to-one correspondence
between the couple (i, j) and the observation number k as follows

site(i) month obs. # at site i(j) obs. # (k)

1 1 1 1

1 2 2 2

1 4 3 3

... ... ... ...

1 18 17 17

2 2 1 18

2 3 2 19

... ... ... ...

Indexing can be very important, especially in models with hierarchies of random
effects.

model {

#The original program contains a model for random recovery rates

#but for simplicity of presentation we consider a constant

#recovery rate equal to the mean empirical recovery rate

#Count part of the model

for (k in 1:N)

{Y[k]~dpois(lambda[k])

lambda[k]<-V[k]*RY[k]*C[k]

log(C[k])<-b[1]*tu[k]+b[2]*te[k]+b[3]*tcoli[k]+b[4]*bas1[k]+

b[5]*bas2[k]+b[6]*bas3[k]+b[7]*ph[k]+

b[8]*logpop[site[k]]+b[9]*logart[site[k]]+

b[10]*lvs[site[k]]

t[k]~dnorm(s[site[k]], taut)

RY[k]<-0.11

}

#Definition of site effects

for (i in 1:M)

{s[i]~dnorm(r[reg[i]],taus)}

#Definition of regional effects

for (j in 1:P)

{r[j]~dnorm(mu,taur)}

#Priors on the coefficients of covariates

for (i in 1:ncovariates)

{b[i]~dnorm(0.0,1.0E-6)}

#Prior on the overall mean

mu~dnorm(0.0,1.0E-6)

#Prior on the precision parameters

taut~ dgamma(0.001,0.001)

taus~ dgamma(0.001,0.001)

taur~ dgamma(0.001,0.001)

#Deterministic transformations. Standard deviations

sigmat<-1/sqrt(taut)

sigmas<-1/sqrt(taus)

sigmar<-1/sqrt(taur)

}

An interesting trick used in WinBUGS is called nested indexing. We used nested
indexing twice in the program, in the definition of time-site and site random effects.
For example, for time-site effects we introduced an additional column vector called
site[], where site[k] is the site corresponding to the k-th observation. Therefore
the BUGS code

for (k in 1:N)

{t[k]~dnorm(s[site[k]], taut)}

specifies that the time-site effect tk is centered around the site effect si corresponding
to observation k. Similarly

for (i in 1:M)

{s[i]~dnorm(r[reg[i]],taus)}

specifies that the time-site effect si is centered around the site effect rg corresponding
to site i. Here reg[i] represents the index of the region corresponding to the i-th
site effect.

Data. Data needed for this program are

a. Y[]: N × 1 dimensional vector of pathogen counts

b. V[]: N × 1 vector of volumes of water analyzed

c. tu[], te[], tcoli[], bas1[], bas2[], bas3[], ph[]: N×1 dimensional vectors
of covariates

d. site[]: N × 1 dimensional vector of site indexes

e. logpop[], logart[], lvs[]: M × 1 dimensional vectors of covariates

f. reg[]: M × 1 dimensional vector of region indexes

g. N, M, and P: numbers of observations, sites and regions respectively.

Note that data in a., b., c., d. can be entered in rectangular format using an
N × 10 matrix. Data in e. and f. can be entered in rectangular format using an
M × 4 matrix. The three constants from point g. can be entered in a list.

Initial values. Initial values should be provided for the following random vari-
ables

a. t[]: the N × 1 dimensional vector of time-site random effects.

b. s[]: the M × 1 dimensional vector of site random effects.

c. r[]: the P × 1 dimensional vector of regional random effects.

d. b[]: the vector of parameters of covariates

e. mu: the overall mean

f. taut, taus, taur: preciosions of time-site, site and regional random effects
respectively.

5 Mixing and Reparameterization

Mixing is the property of the MCMC to move rapidly throughout the support of the
posterior distribution of parameters given the data. Finding a good candidate model
can be a very computationally intensive process of exploring many models. Therefore,
reducing computational time for MCMC simulations could be critical.

The within-chain correlation for MCMC samples determines the required number
of MCMC simulations to achieve a target accuracy. Strong posterior correlation
among parameters was found to be the main reason for high within-chain correlation
for MCMC sampling (Gilks and Roberts 1996). We explore methods of reducing these
posterior correlations through simple techniques that can easily be implemented for
a wide range of models.

There exist non-standard sampling algorithms, random and adaptive direction
sampling, simulated tempering, Metropolis-coupled MCMC, and data augmentation,
to reduce within-chain correlation without reducing posterior correlation among pa-
rameters (Gilks and Roberts 1996; Robert and Casella, 1999). However their net
improvement in mixing for complex models is not clearly understood and their ap-
plicability limited by lack of software. For standard MCMC simulations WinBugs is
the available tool.

We will use the Effective Sample Size (ESS) as a measure of mixing. ESS for a
parameter is the sample size of an independent sample giving the same estimation

accuracy as the correlated MCMC sample. If K is the total number of correlated
MCMC simulations and ρk is the lag k autocorrelation for one parameter, then

ESS =
K

1 + 2
∑∞

k=1 ρk

. (5)

ESS can be estimated using the sample autocorrelation function (ACF).
Improving mixing of MCMC simulation is model specific and solutions are often a

mixture of art and science. Therefore, we seek simple solutions for the lack of efficiency
of MCMC that can easily be implemented for a variety of models and situations.

A critical feature of the ICR data set is the large number of zero counts (≥
90%) which carry relatively little information about the parameters. For example,
500 counts that are all zero are consistent with a mean of 10−4 as well as 10−10.
However, for assessing health risk, a mean of 10−4 has rather different implications
than a mean of 10−10, so distinguishing between them is a concern. This lack of
information appears also to cause poor mixing. To investigate this problem consider
a test example.

5.1 Test example

The following model was used to simulate sets of counts

Yi |λi ∼ Poisson(λi)
log(λi) = ti
ti |µ, σt ∼ N(µ, σ2

t )





(6)

We generated samples of size n = 200 with σ = 1 for µ = −2, −1, 0 and 1 correspond-
ing to unconditional Poisson means for λi of 0.22, 0.61, 1.65, 4.48. In this framework
the information in the sample is represented by the mean count and the percent of
non-zero counts. Table 1 provides a summary of these information measures for 4
simulated samples, one for each combination of parameters.

Since σ is fixed, as µ increases one sees an increase in the sample mean and
decrease in the percent of zero counts. We see in Table 1 as µ increases, the samples
become more and more informative about µ and σ because the posterior standard
deviations of both µ and σ decrease.

MCMC simulations using WinBugs 1.4 are used for Bayesian statistical inference
for model (6) for the samples considered. To compare the ESS values one needs
to assess their variability due to ACF estimation. For each simulated data set cor-
responding to a set of parameters we ran 10 independent MCMC simulations and
estimated ESS for each run. Table 2 presents the average and the standard errors of
estimated ESS for these samples.

In Table 2 as we move from less informative samples to more informative samples,
ESS increases dramatically, suggesting that improving mixing is especially important
when the individual observations contain little information, as in the ICR data.

Table 1: Measures of information in the samples

µ = −2 µ = −1 µ = 0 µ = 1

Sample mean 0.22 0.48 1.58 3.90
% zero counts 84 67 35 18
SD(µ) 0.43 0.21 0.11 0.08
SD(σ) 0.47 0.24 0.11 0.08
E(µ) −1.92 −1.07 0.03 0.97
E(σ) 0.66 0.80 0.91 0.91

Notes: Total number of MCMC simulations K=10,000. Entries are the values for
three independent samples. SD is the posterior standard deviation and E is the
posterior mean.

Table 2: ESS for posterior simulations of length 10,000

µ = −2 µ = −1 µ = 0 µ = 1

ESSµ 110 221 1569 3140

ESSσ 61 179 1391 2229

SE(ESSµ) 6.7 16.1 57.4 110.8

SE(ESSσ) 4.6 16.0 63.3 79.2

Notes: ESSµ and ESSσ are the averages of estimated ESS from 10 independent simu-

lations of the posterior distribution of µ and σ respectively. SE(ESSµ) and SE(ESSσ)

are standard errors of the means ESSµ and ESSσ.

5.2 Improving Mixing Strategies

We will focus on several simple strategies of improving mixing: centering and or-
thogonalization of covariates, hierarchical centering of random effects and analytical
integration of random effects.

Centering covariates is widely used to make them orthogonal to the intercept.
A more complex method is orthogonalization of covariates. If X is the matrix of
covariates then we want an invertible matrix A such that Z = XA is orthogonal.
Thus, we let A be the inverse of a symmetric square-root of X ′X. If β is the coefficient

vector in the X parameterization, then β∗ = A−1β is the coefficient vector in the Z
parameterization since Xβ = Zβ∗. In the MCMC iterations we use the β∗ parameters,
but we can also monitor the original parameters β = Aβ∗ and use them later for
inference.

Hierarchical centering is a simple reparameterization technique for models with
several layers of random effects. Model (4) is presented with random effects hierar-
chically centered. An equivalent form of that model without hierarchical centering of
random effects is

Yij |λij ∼ Poisson(λij)
λij = Vij Rij Cij

log(Cij) = X ′
ijβ + tij + sj + rk(j) + µ

tij ∼ N(0, σ2
t )

sj ∼ N(0, σ2
s)

rk(j) ∼ N(0, σ2
r).





(7)

In the Bayesian analysis of ANOVA models, hierarchical centering can significantly
improve of the posterior correlation structure (Gilks, et al., 1996).

Table 3 presents the ESS for different combinations of reparameterizations for
model (4). Because estimates of ESS have their own variability due to ACF esti-
mation, comparing two different ESS may not be very informative when they are
close. However ESS values that differ by a factor of 2 are likely to correspond to real
differences in mixing rates.

A first observation is that ESS is much smaller than the MCMC sample size. No
clear pattern can be observed for all parameters. However, for the average regional
mean µ, Turbidity, Region 8 and Season 1 the combination HC+C+O greatly im-
proves mixing. For each variable in Table 3 we can decide what reparameterization
is more suitable.

One can use a max-min strategy that maximizes the smallest ESS over all param-
eters of interest. In our application the reparameterizations maximizing the smallest
ESS are HC+C and HC+C+O. Because the first reparameterization is simpler it was
chosen for future analysis.

Consider the following two equivalent models

Yij |λij ∼ Poisson(λij)
λij = Vij Rij Cij

Cij = exp(X ′
ijβ + sj) gij

gij |α ∼ Gamma(α, α)
sj |rk(j), σs ∼ N(rk(j), σ

2
s)

rk |µ, σr ∼ N(µ, σ2
r).





(8)

Table 3: Estimated ESS for different model parameterizations

Variable Standard HC HC + C C + O HC + C + O

µ 15 30 200 30 120
Turbidity 85 45 230 560 225
Reservoir 125 65 110 190 110
Region 8 15 75 1160 30 890
Season 1 130 115 290 875 815
σr 95 1130 1160 195 1200
σs 205 90 110 310 100
σt 95 90 55 110 85

Notes: Total number of MCMC simulations K=10,000. HC=Hierarchical centering,
C=Centering, O=Orthogonalization

and
Yij |λij ∼ Negative Binomial

(
α

α+λij
, α

)

λij = Vij Rij exp(X ′
ijβ + sj)

Rij |a, b ∼ Beta(a, b)
sj |rk(j), σs ∼ N(rk(j), σ

2
s)

rk |µ, σr ∼ N(µ, σ2
r).





(9)

Analytical integration of random effects reduces the dimension of the posterior
distribution and can improve ESS. For examples and details see Crainiceanu et al.,
2002.

6 Script language in WinBUGS

The scripting language is an alternative to the menu/dialog box interface of Win-
BUGS. This language can be useful for automating routine analysis. The language
works by writing values into fields and clicking on buttons.

A minimum of four files are required:

a. The script file itself. This file has to be in WinBUGS format (.odc)

b. A file containing the BUGS language representation of the model (.txt)

c. A file (or several) containing the data (.txt)

d. A file for each chain containing initial values (.txt).

The shortcut BackBUGS has been set up to run the commands contained in the
file script.odc (in the root directory of WinBUGS) when it is double-clicked. A

WinBUGS session may be embedded within any software component that can execute
the BackBUGS shortcut.

A list of currently implemented commands in the scripting language can be found
in the BUGS manual.

We provide an example of script file that works

display(’log’) #Choose display type

check(’Test1/pathogen.txt’) #Model check

data(’Test1/pathogen_dat.txt’) #Load data

compile(1) #Compile model

inits(1, ’Test1/pathogen_in.txt’) #Initialize parameters

gen.inits() #Generate initial values

update(4000) #4000 burn-in iterations

set(b[]) #Set b to be monitored

set(mu)

set(taut)

set(taus)

set(taur)

update(10000) #Do 10000 iterations

stats(*) #Provide stats for all parameters

history(*) #Provide chain histories

trace(*) #Provide chain traces

density(*) #Provide kernel density estimates

autoC(*) #Provide autocorrelations

quantiles(*) #Provide running quantiles

coda(*,output) #Save the two files for coda

save(’pathogenLog’) #Save the output file

7 Homework

The ultimate goal of this homework is to construct a WinBUGS program for inference
on a nonparametric Bayesian model. A few concepts needed for doing this homework
are introduced below.

7.1 Penalized splines as GLMMs

Consider the partially linear model yi = W T
i γ + m (xi) + εi , where εi are i.i.d.

N (0, σ2
ε ), εi is independent of Wi and xi, Wi is a vector of covariates that enter the

model linearly, xi is another covariate, and m(·) is a smooth function.
We model m(·) using the class of spline functions

m (x, θ) = β0 + β1x + . . . + βpx
p +

K∑

k=1

bk (x− κk)
p
+ ,

where θ = (β0, . . . , βp, b1, . . . , bK)T is the vector of regression coefficients, and κ1 <
κ2 < . . . < κK are fixed knots. Here a+ is equal to a if a > 0 and zero otherwise and
ap

+ = (a+)p. Ruppert, Wand and Carroll (2003) provide a comprehensive framework
for semiparametric smoothing using splines, Following Ruppert (2002), we consider a
number of knots that is large enough (typically 5 to 20) to ensure the desired flexibility,
and κk is the sample quantile of x’s corresponding to probability k/(K + 1), but our
results hold for any other choice of knots. To avoid overfitting, we minimize

n∑

i=1

{
yi −W T

i γ −m (xi, θ)
}2

+
1

λ
θT Dθ , (10)

where λ is the smoothing parameter and D is a known positive semi-definite penalty
matrix. We focus on matrices D of the form

D =

[
0(p+1)×(p+1) 0(p+1)×K

0K×(p+1) Σ−1

]
,

where Σ is a positive definite matrix and 0m×l is an m× l matrix of zeros. This matrix
D penalizes the coefficients of the spline basis functions (x− κk)

p
+ only. We will use

Σ = IK .
Let Y = (y1, y2, . . . , yn)T , W be the matrix with the i-th row equal to W T

i , X be
the matrix with the i-th row Xi = (1, xi, . . . , x

p
i ), and Z be the matrix with i-th row

Zi =
{
(xi − κ1)

p
+ , . . . , (xi − κK)p

+

}
.

If we divide (10) by the error variance one obtains

1

σ2
ε

‖Y −Wγ −Xβ − Zb‖2 +
1

λσ2
ε

bT Σ−1b ,

where β = (β0, . . . , βp)
T and b = (b1, . . . , bK)T . Define σ2

b = λσ2
ε , consider the vectors

γ and β as fixed parameters and the vector b as a set of random parameters with
E(b) = 0 and cov(b) = σ2

bΣ. If (bT , εT )T is a normal random vector and b and ε are
independent then one obtains an equivalent model representation of the penalized
spline in the form of a linear mixed model (Brumback et al., 1999):

Y = Wγ + Xβ + Zb + ε, cov

(
b
ε

)
=

[
σ2

bΣ 0
0 σ2

ε In

]
. (11)

For this model E(Y ) = Wγ + Xβ and cov(Y ) = σ2
ε Vλ, where Vλ = In + λZΣZT and

n is the total number of observations. In the LMM described in (11) both W and X
correspond to fixed effects. For simplicity of presentation we can redefine X = [W |X],
β = (γT , βT )T and let p + 1 be the dimension of the new vector β.

Note that model (11) can easily be implemented in WinBUGS. For example the
likelihood part of the model can be implemented as

for (i in 1:n){

response[i]~dnorm(m[i],taueps)

m[i]<-inprod(beta[],X[i,])+inprod(b[],Z[i,])}

where X[,] is the design matrix for fixed effects (in this case additional covariates and
monomials) and Z[,] is the design matrix for random effects. These matrices can be
obtained outside WinBUGS and then loaded as data into WinBUGS. Matrices can
also be manipulated within WinBUGS. The function inprod(,) is the inner product
of two vectors.

Just as in the case of ridge regression presented in section 4.2 the specification of
the LMM is completed by describing the distribution of the b parameters

for (k in 1:K){b[k]~dnorm(0,taub)}

The model also includes vague priors on all hyperparameters

for (l in 1:p+1){beta[l]~dnorm(0,1.0E-6)}

taueps~dgamma(1.0E-3,1.0E-3)

taub~dgamma(1.0E-3,1.0E-3)

7.2 The Wage–Union Membership Data

In this subsection the actual homework problem is described. Download from the
website the wageunion.txt file containing data on wages/h and union membership
for 534 workers. The data were taken from the Statlib website at Carnegie Mellon
University lib.stat.cmu.edu.

Denote by yi the union membership of the i-th worker and by xi his wage/h. We
can use a Binomial/Logit model to link yi and xi as follows

{
yi | pi ∼ Binomial(pi, 1)
logit(pi) = m(xi)

(12)

Problem 1. Write WinBUGS programs for the following cases





m(x) = β0 + β1x
m(x) = β0 + β1x + β2x

2

m(x) = β0 + β1x + β2x
2 + β3x

3

(13)

a. Provide the WinBUGS program for the first case and discuss code changes for
the other programs. Shorter programs are encouraged.

b. Plot the data and the posterior means for each of the three mean functions on
the same graph.

c. Discuss your results

Problem 2. Suppose now that the logit of p(x) is modeled nonparametrically
using a linear (p = 1) penalized spline with K = 20 knots. Choose the knots at the
sample quantiles of x’s corresponding to probabilities 1/21, 2/21, . . . , 20/21. We use
the following model





yi|xi ∼ Binomial{p(xi), 1}
logit{p(xi)} = β0 + β1xi +

∑K
k=1 bk(xi − κk)+

bk ∼ N(0, σ2
b )

εi ∼ N(0, σ2
ε )

, (14)

a. Provide the WinBUGS program. The shorter, the better. Hint: use coding
ideas presented in Section 7.1.

b. Plot the posterior means and the 95% credible intervals for the mean function.

c. Discuss your results.

d. Discuss mixing properties.

